Beauville Periods in OEIS

Periods Converter

See also: https://community.wolfram.com/groups/-/m/t/2926119

In[1]:=
```
$WPFCs = {...}
```

Out[1]=
```
{-21 G2[0] G2[1]^3 G3[0] + 18 G2[0]^2 G2[1]^2 G3[1] +
  8 G2[0]^3 G2[2] × G3[1] - 216 G2[2] G3[0]^2 G3[1] + 108 G2[1] × G3[0] G3[1]^2 -
  120 G2[0] G3[1]^3 - 8 G2[0]^3 G2[1] × G3[2] + 216 G2[1] G3[0]^2 G3[2],
 -16 (9 G2[0]^2 G2[1]^2 G3[0] - 3 G2[0]^3 G2[2] × G3[0] +
    81 G2[2] G3[0]^3 - 7 G2[0]^3 G2[1] × G3[1] - 135 G2[1] G3[0]^2 G3[1] +
    108 G2[0] × G3[0] G3[1]^2 + 2 G2[0]^4 G3[2] - 54 G2[0] G3[0]^2 G3[2]),
 16 (G2[0]^3 - 27 G3[0]^2) (-3 G2[1] × G3[0] + 2 G2[0] × G3[1])}
```

In[3]:=
```
EllipticPeriodsAnnihilator[
  g2x_, g3x_] := With[{res0 = ReplaceAll[
      $WPFCs, {
       G2[n_] :> D[g2x, {x, n}],
       G3[n_] :> D[g3x, {x, n}]
      }
     ]},
  Factor[Cancel[Divide[#, PolynomialGCD @@ #
       ] &[Together[res0]]
   ]]
 ]
```

After Herfurtner 1991

https://link.springer.com/article/10.1007/BF01445211

1,1,1,9 * ~ A006077 ~ BI, BVI

$I_1 I_1 I_1 I_9$

$(1, \eta, \eta^2, \infty)$

$$G_2 = 3X(9X^3 - 8Y^3)$$
$$G_3 = 27X^6 - 36X^3Y^3 + 8Y^6$$
$$\Delta = 2^6 \cdot 3^3 Y^9 (X^3 - Y^3)$$
$$\mathcal{J} = \frac{1}{64} \frac{X^3(9X^3 - 8Y^3)^3}{Y^9(X^3 - Y^3)}$$
$$\mathrm{CR} = -\eta, \quad \eta = e^{\frac{2\pi i}{3}}$$

In[12]:=
```
data = {
    3 x (9 x^3 - 8 y^3),
    27 x^6 - 36 x^3 y^3 + 8 y^6
   } /. y → 1 /. x → -9 x + 1
```
Out[12]=
```
{3 (-8 + 9 (1 - 9 x)^3) (1 - 9 x), 8 - 36 (1 - 9 x)^3 + 27 (1 - 9 x)^6}
```

In[13]:=
```
ode = EllipticPeriodsAnnihilator @@ data
```
Out[13]=
```
{3 (-1 + 9 x), (-1 + 9 x)^2, x (1 - 9 x + 27 x^2)}
```

proof:

In[14]:=
```
gf = 1 / (1 - 3 x) Hypergeometric2F1[1 / 3, 2 / 3, 1, ((-3 x) / (1 - 3 x)) ^3];
```
In[15]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```
Out[15]=
```
0
```

In[16]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```
Out[16]=
```
{1, 3, 9, 21, 9, -297, -2421, -12933, -52407, -145293, -35091}
```

Another proof:

```
ode = Factor[9 hypergeometric /. {a → 1 / 3, b → 2 / 3, c → 1}] == 0
ode2 = ODEChangeVariables[ode, {T}, x, z → ((-3 x) / (1 - 3 x)) ^3]["Result"];
ode3 = ode2 /. Subscript[T, 1] → Function[z, (1 - 3 z) G[z]];
final =
 Divide[Factor[Coefficient[-ode3[[1]], D[G[x], {x, #}]] & /@ {0, 1, 2}], (3 x - 1) ^3]
```

Out[ ]=

$2\,T[z] - 9\,T'[z] + 18\,z\,T'[z] - 9\,z\,T''[z] + 9\,z^2\,T''[z] == 0$

Out[ ]=

$\{3\,(-1 + 9\,x),\ (-1 + 9\,x)^2,\ x\,(1 - 9\,x + 27\,x^2)\}$

1,1,2,8 * ~ A081085 ~ BII, BV

$I_1 I_1 I_2 I_8$ $(-1, 1, 0, \infty)$

$$G_2 = 3(16X^4 - 16X^2Y^2 + Y^4)$$
$$G_3 = 64X^6 - 96X^4Y^2 + 30X^2Y^4 + Y^6$$
$$\Delta = 2^2 \cdot 3^6 X^2 Y^8 (X+Y)(X-Y)$$
$$\mathcal{J} = \frac{1}{108}\,\frac{(16X^4 - 16X^2Y^2 + Y^4)^3}{X^2Y^8(X+Y)(X-Y)}$$
$$\mathrm{CR} = -1$$

In[19]:=

```
data = {
    3 (16 x^4 - 16 x^2 y^2 + y^4),
    64 x^6 - 96 x^4 y^2 + 30 x^2 y^4 + y^6
   } /. y → 1 /. x → -8 x + 1
```

Out[19]=

$\{3\,(1 - 16\,(1 - 8\,x)^2 + 16\,(1 - 8\,x)^4),\ 1 + 30\,(1 - 8\,x)^2 - 96\,(1 - 8\,x)^4 + 64\,(1 - 8\,x)^6\}$

In[21]:=

```
ode = EllipticPeriodsAnnihilator @@ data
```

Out[21]=

$\{4\,(-1 + 8\,x),\ 1 - 24\,x + 96\,x^2,\ x\,(-1 + 4\,x)\,(-1 + 8\,x)\}$

proof:

In[22]:=

$gf = 4 / Pi / 2\ \mathrm{EllipticK}\left[\frac{16\,x^2}{(1 - 4\,x)^2}\right] \Big/ (1 - 4\,x)$

Out[22]=

$$\frac{2\,\mathrm{EllipticK}\left[\frac{16\,x^2}{(1-4\,x)^2}\right]}{\pi\,(1 - 4\,x)}$$

In[23]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```
Out[23]=
```
0
```

In[24]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```
Out[24]=
```
{1, 4, 20, 112, 676, 4304, 28 496, 194 240, 1 353 508, 9 593 104, 68 906 320}
```

1,2,3,6 * ~ A002893, A093388, A000172 ~BIV

$I_1 I_2 I_3 I_6$ $(4, -\tfrac{1}{2}, 0, \infty)$

$$G_2 = 12(X^4 - 4X^3Y + 2XY^3 + Y^4)$$
$$G_3 = 4(2X^6 - 12X^5Y + 12X^4Y^2 + 14X^3Y^3 + 3X^2Y^4 + 6XY^5 + 2Y^6)$$
$$\Delta = 2^4 \cdot 3^6 X^3 Y^6 (2X+Y)^2 (X-4Y)$$
$$\mathcal{J} = \frac{4}{27} \frac{(X^4 - 4X^3Y + 2XY^3 + Y^4)^3}{X^3Y^6(2X+Y)^2(X-4Y)}$$
$$\mathrm{CR} = -8$$

A002893

In[29]:=
```
data = Factor[{
    12 * (x^4 - 4 x^3 y + 2 x y^3 + y^4),
    4 (2 x^6 - 12 x^5 y + 12 x^4 y^2 + 14 x^3 y^3 + 3 x^2 y^4 + 6 x y^5 + 2 y^6)
   } /. y → 1 / 4 + 1 / 4 x /. x → -x]
```
Out[29]=

$$\left\{\frac{3}{64}(1+3x)\left(1-15x+75x^2+3x^3\right), -\frac{1}{512}\left(-1+6x+3x^2\right)\left(1-12x+30x^2-540x^3+9x^4\right)\right\}$$

In[30]:=
```
ode = EllipticPeriodsAnnihilator @@ data
```
Out[30]=

$$\left\{-3(-1+3x), -1+20x-27x^2, -((-1+x)x(-1+9x))\right\}$$

In[31]:=

$$\text{gf} = -\frac{4\,\text{EllipticK}\left[\frac{\left(\sqrt{1+\left(-6+8\sqrt{x}-3x\right)x}-\sqrt{1-x\left(6+8\sqrt{x}+3x\right)}\right)^2}{\left(\sqrt{1+\left(-6+8\sqrt{x}-3x\right)x}+\sqrt{1-x\left(6+8\sqrt{x}+3x\right)}\right)^2}\right]}{-\pi\sqrt{-\left(1+\sqrt{x}\right)^3\left(-1+3\sqrt{x}\right)}-\pi\sqrt{-\left(-1+\sqrt{x}\right)^3\left(1+3\sqrt{x}\right)}};$$

In[32]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```
Out[32]=
```
0
```

In[33]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```
Out[33]=
```
{1, 3, 15, 93, 639, 4653, 35 169, 272 835, 2 157 759, 17 319 837, 140 668 065}
```

A093388

In[34]:=
```
data = Factor[{
    12 * (x^4 - 4 x^3 y + 2 x y^3 + y^4),
    4 (2 x^6 - 12 x^5 y + 12 x^4 y^2 + 14 x^3 y^3 + 3 x^2 y^4 + 6 x y^5 + 2 y^6)
   } /. y → 1 / 4 + 1 / 4 x /. x → -x /. x → -8 x + 1]
```

Out[34]=

$$\{12\,(-1+6\,x)\,\left(-1+18\,x-84\,x^2+24\,x^3\right),\ -8\,\left(1-12\,x+24\,x^2\right)\,\left(-1+24\,x-192\,x^2+504\,x^3+72\,x^4\right)\}$$

In[35]:=
```
ode = EllipticPeriodsAnnihilator @@ data
```

Out[35]=

$$\{-6\,(-1+12\,x),\ -1+34\,x-216\,x^2,\ -x\,(-1+8\,x)\,(-1+9\,x)\}$$

In[36]:=
```
gf = Hypergeometric2F1[1 / 3, 2 / 3, 1, x^2 * (8 * x - 1) / (2 * x - 1 / 3) ^3] / (1 - 6 * x);
```

In[37]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```

Out[37]=

0

In[38]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```

Out[38]=

{1, 6, 42, 312, 2394, 18 756, 149 136, 1 199 232, 9 729 882, 79 527 084, 654 089 292}

A000172

In[39]:=
```
data = Together[{
    12 * (x^4 - 4 x^3 y + 2 x y^3 + y^4),
    4 (2 x^6 - 12 x^5 y + 12 x^4 y^2 + 14 x^3 y^3 + 3 x^2 y^4 + 6 x y^5 + 2 y^6)
   } /. y → 1 / 4 + 1 / 4 x /. x → -x /. x → -8 / 9 x + 1 / 9]
```

Out[39]=

$$\left\{\frac{4}{243}\,\left(1-8\,x+240\,x^2-464\,x^3+16\,x^4\right),\ -\frac{8\,\left(1-12\,x-480\,x^2+3080\,x^3-12\,072\,x^4+4128\,x^5+64\,x^6\right)}{19\,683}\right\}$$

In[40]:=
```
ode = EllipticPeriodsAnnihilator @@ data
```

Out[40]=

$$\{-2\,(1+4\,x),\ 1-14\,x-24\,x^2,\ -x\,(1+x)\,(-1+8\,x)\}$$

In[41]:=
```
gf = Hypergeometric2F1[1 / 3, 2 / 3, 1, 27 x^2 / (1 - 2 x) ^3] / (1 - 2 x);
```

In[42]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```

Out[42]=

0

In[43]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```

Out[43]=

{1, 2, 10, 56, 346, 2252, 15 184, 104 960, 739 162, 5 280 932, 38 165 260}

1,1,5,5 ~ A005258 ~ BIII

$I_1 I_1 I_5 I_5$ $(\omega_1, \omega_2, 0, \infty)$

$$G_2 = 3(X^4 - 12X^3Y + 14X^2Y^2 + 12XY^3 + Y^4)$$
$$G_3 = X^6 - 18X^5Y + 75X^4Y^2 + 75X^2Y^4 + 18XY^5 + Y^6$$
$$\Delta = 2^6 \cdot 3^6 X^5 Y^5 (X^2 - 11XY - Y^2)$$
$$\mathscr{J} = \frac{1}{2^6 \cdot 3^3} \frac{(X^4 - 12X^3Y + 14X^2Y^2 + 12XY^3 + Y^4)^3}{X^5Y^5(X^2 - 11XY - Y^2)}$$
$$\mathrm{CR} = \left(\frac{1+\sqrt{5}}{1-\sqrt{5}}\right)^5, \quad \omega_{1,2} = \left(\frac{1 \pm \sqrt{5}}{2}\right)^5$$

In[44]:=
```
data = {
    3 * (x^4 - 12 x^3 * y + 14 x^2 y^2 + 12 x * y^3 + y^4),
    x^6 - 18 x^5 * y + 75 * x^4 * y^2 + 75 * x^2 y^4 + 18 * x * y^5 + y^6
   } /. y → 1 /. x → -x
```

Out[44]=

$\{3 (1 - 12 x + 14 x^2 + 12 x^3 + x^4), 1 - 18 x + 75 x^2 + 75 x^4 + 18 x^5 + x^6\}$

In[45]:=
```
ode = EllipticPeriodsAnnihilator @@ data
Factor[CoefficientList[ode[[1]], x]]
Factor[CoefficientList[D[ode[[3]], x] - ode[[2]], x]]
```

Out[45]=

$\{3 + x, -1 + 22 x + 3 x^2, x (-1 + 11 x + x^2)\}$

Out[46]=

{3, 1}

Out[47]=

{}

In[48]:=
```
gf = Divide[Hypergeometric2F1[1 / 12, 5 / 12, 1,
    1728 * x^5 * (1 - 11 * x - x^2) / (1 - 12 * x + 14 * x^2 + 12 * x^3 + x^4)^3
   ], (1 - 12 * x + 14 * x^2 + 12 * x^3 + x^4)^(1 / 4)]
```

Out[48]=

$$\frac{\text{Hypergeometric2F1}\left[\frac{1}{12}, \frac{5}{12}, 1, \frac{1728 x^5 (1 - 11 x - x^2)}{(1 - 12 x + 14 x^2 + 12 x^3 + x^4)^3}\right]}{(1 - 12 x + 14 x^2 + 12 x^3 + x^4)^{1/4}}$$

In[49]:=
```
FullSimplify[Dot[ode, D[gf, {x, #}] & /@ {0, 1, 2}]]
```

Out[49]=

0

In[50]:=
```
CoefficientList[Series[gf, {x, 0, 10}], x]
```

Out[50]=

{1, 3, 19, 147, 1251, 11 253, 104 959, 1 004 307, 9 793 891, 96 918 753, 970 336 269}

Acknowledgements

Many thanks to Duco Van Straten for bringing this reference to our attention.

Author

Bradley Klee

May 29, 2023

See also

Don Zagier, “Integral solutions...” (2009).

https://people.mpim-bonn.mpg.de/zagier/files/tex/AperylikeRecEqs/fulltext.pdf

$$\text{Sequence } \mathbf{A}: \quad u_n = \sum_{k=0}^{n} \binom{n}{k}^3 = \sum_{k=0}^{n} \binom{n}{k}^2 \binom{2k}{n},$$

$$\text{Sequence } \mathbf{B}: \quad u_n = \sum_{k=0}^{[n/3]} (-1)^k \, 3^{n-3k} \binom{n}{3k} \binom{3k}{k} \binom{2k}{k},$$

$$\text{Sequence } \mathbf{C}: \quad u_n = \sum_{k=0}^{n} \binom{n}{k}^2 \binom{2k}{k},$$

$$\text{Sequence } \mathbf{D}: \quad u_n = \sum_{k=0}^{n} \binom{n}{k}^2 \binom{n+k}{k} = \sum_{k=0}^{n} (-1)^{n-k} \binom{n}{k} \binom{n+k}{k}^2,$$

$$\text{Sequence } \mathbf{E}: \quad u_n = \sum_{k=0}^{[n/2]} 4^{n-2k} \binom{n}{2k} \binom{2k}{k}^2.$$

The formulas for the sequences **B** and **E** say that the corresponding generating functions $\sum u_n t^n$ have hypergeometric representations:

$$F_{\mathbf{B}}(t) = \frac{1}{1-3t} F\Big(\frac{1}{3}, \frac{2}{3}; 1; \big(\frac{-3t}{1-3t}\big)^3\Big), \qquad F_{\mathbf{E}}(t) = \frac{1}{1-4t} F\Big(\frac{1}{2}, \frac{1}{2}; 1; \big(\frac{4t}{1-4t}\big)^2\Big).$$

nor (since the elliptic surfaces have non-trivial automorphisms) need they all be different. The actual correspondence turns out to be as follows:

Beauville Family:	IV	I, VI	III	II, V
Picard-Fuchs equation:	**A**, **C**, **F**	**B**, #14	**D**	**E**, **G**, #11

One way to see this is to compute the j-invariant of each of Beauville's family as a rational function of t, invert this equation to compute t as an explicit modular function, and compare with the list of t's which we found before. The j-invariant in turn is most conveniently found by rewriting Beauville's equations in Weierstrass form $y^2 = f(x)$ (f a polynomial of degree 3) . For the reader's convenience, we give the transformation needed in each of the 6 cases do this, and the resulting formula for the j-invariant:

	Transformation to Weierstrass form	j-invariant
I	$X, Y = t(x \pm y),\ Z = 2(1+3x)$	$-t^3(t^3-216)^3/(t^3+27)^3$
II	$X = xt,\ Y = x+y,\ Z = t$	$2^8(t^4-t^2+1)^3/t^4(t^2-1)^2$
III	$X = 1+(y+tx)/(1-x),\ Y = 2x,\ Z = 2$	$(t^4-12t^3+14t^2+12t+1)^3/t^5(t^2-11t-1)$
IV	$X, Y = 1+tx \pm y,\ Z = 2x(1+tx)$	$(t+2)^3(t^3+6t^2-12t+8)^3/t^3(t-1)^2(t+8)$
V	$X, Y = 1+tx \pm y,\ Z = 2x(1+tx)$	$(t^4+16t^2+16)^3/t^2(t^2+16)$
VI	$X = 2x,\ Y = y-xt-1,\ Z = 2x^2$	$-t^3(t^3+24)^3/(t^3+27)$